30/replication_package/Dofiles/002_prepare_data_full.do

* LAST CHANGED: DP March 2021

clear

set mem 500m
set more off
set logtype text
set matsize 100

cap log close
log using "$Log/002_prepare_data_full.log", replace


* Outline:  
* (1) Use the clean data, get the data into shape for analysis

* (1a) Use the Clean Cosponsor Data, add to it individual and district characteristics
use "$RawData/Cosponsors_Clean.dta", clear
sort v2 state_abbrev district term_served
merge v2 state_abbrev district term_served using "$RawData/IndividualCharacteristics_Clean.dta", /*
*/    keep(v2 state_abbrev state_icpsr district term_served name_clean v1_fix female party age tenure_run pus comc comr dwnom1 dwnom2 /*
*/         borninstate agestart privatesec anycoll namedcoll ivycoll statecoll anygrad jd mba phd military /*
*/         occ0 occ1 occ2 occ3 occ4 occ5 occ6 DSpndPct DDonaPct CQRating3 CQRating3_01 black native asian latino)
tab _merge
drop if _merge==2
drop _merge

gen DSpndPct_miss=DSpndPct==.
replace DSpndPct=50 if DSpndPct==.

sort v2 state_abbrev district term_served
merge v2 state_abbrev district term_served using "$RawData/Committee_Clean.dta", /*
*/    keep(c_*)
tab _merge
drop _merge

sort v2 state_abbrev district
merge v2 state_abbrev district using "$RawData/DistrictCharacteristics_Clean.dta", /*
*/    keep(v2 state_abbrev district losing_candidate_clean_cqq losing_candidate_gender mixed_gender_election /*
*/         sample1 demshare1 repshare1 MV1_democrat tot_pop pct_age_over65 pct_black pct_for_born pct_urban med_inc_all lnpop lnarea lninc /*
*/         charisma_dem predicted_dem rep_incumbent_cqq dem_incumbent_cqq minor_incumbent_cqq lag_demshare1)
tab _merge
drop _merge

foreach var of varlist state_abbrev state_icpsr district term_served name_clean v1_fix female party age tenure_run pus comc comr dwnom1 dwnom2  /*
*/                     losing_candidate_clean_cqq losing_candidate_gender mixed_gender_election /* 
*/                     sample1 demshare1 repshare1 MV1_democrat charisma_dem predicted_dem /*
*/                     tot_pop pct_age_over65 pct_black pct_for_born pct_urban med_inc_all lnpop lnarea lninc /*
*/                     rep_incumbent_cqq dem_incumbent_cqq minor_incumbent_cqq lag_demshare1 /*
*/                     borninstate agestart privatesec anycoll namedcoll ivycoll statecoll anygrad jd mba phd military /*
*/                     occ0 occ1 occ2 occ3 occ4 occ5 occ6 DSpndPct DSpndPct_miss DDonaPct CQRating3 CQRating3_01 c_* black native asian latino {
    ren `var' cosp_`var'
}

foreach name in "state_abbrev" "state_icpsr" "district" "term_served" "v1_fix" "age" "female" "tenure_run" "party" {
   ren cosp_`name' cosponsor_`name'    // this to be consistent with previous code
}

egen cosponsor_v1_flex=group(cosponsor_v1_fix v2)

sort v2 HRnumber
save "$IntermediateData/CosponsorsWithInfo_Clean_v2.dta", replace



* (1b) Use the Clean Bill data, add to it individual and district characteristics
clear
use "$RawData/BillCharacteristics_Clean.dta", clear
sort v2 state_abbrev district term_served
merge v2 state_abbrev district term_served using "$RawData/IndividualCharacteristics_Clean.dta", /* 
*/    keep(v2 state_abbrev state_icpsr district term_served name_clean v1_fix female party age pus tenure_run comc comr dwnom1 dwnom2 /*
*/         borninstate agestart privatesec anycoll namedcoll ivycoll statecoll anygrad jd mba phd military /*
*/         occ0 occ1 occ2 occ3 occ4 occ5 occ6 DSpndPct DDonaPct CQRating3 CQRating3_01 black native asian latino)
tab _merge
drop if _merge==2
drop _merge

merge m:1 v2 state_abbrev district term_served using "$RawData/PrimaryElections.dta", keepusing(v2 state_abbrev district term_served mixed_gender_primary MVprim_female)
compress

drop if _merge==2
drop _merge


gen DSpndPct_miss=DSpndPct==.
replace DSpndPct=50 if DSpndPct==.

sort v2 state_abbrev district term_served
merge v2 state_abbrev district term_served using "$RawData/Committee_Clean.dta", /*
*/    keep(c_*)
tab _merge
drop _merge

sort v2 state_abbrev district
merge v2 state_abbrev district using "$RawData/DistrictCharacteristics_Clean.dta", /*
*/    keep(v2 state_abbrev district sample1 republican democratic minor1_vote /*
*/         demshare1 repshare1 MV1_democrat mixed_gender_election /*
*/         losing_candidate_clean_cqq losing_candidate_gender /*
*/         tot_pop pct_age_over65 pct_black pct_for_born pct_urban med_inc_all lnpop lnarea lninc/*
*/         charisma_dem predicted_dem rep_incumbent_cqq dem_incumbent_cqq minor_incumbent_cqq lag_demshare1)
tab _merge 
drop _merge

*******************************************************************************
* IMPORTANT ADDITION (DANIELE, 2013/06/04): Keep only non-private bills   
count
keep if private==0
count
*******************************************************************************


keep v2 HRnumber state_abbrev state_icpsr district term_served name_clean sponsor v1_fix intro_date /*
*/   numb_cosponsors numb_committees_master plaw_master lma_date iter_length mult commem house_* /*
*/   major minor passh passs plaw_cbp plawdate female party age  pus tenure_run /*
*/   comc comr dwnom1 dwnom2 republican democratic minor1_vote losing_candidate_clean_cqq /*
*/   losing_candidate_gender mixed_gender_election sample1 demshare1 repshare1 MV1_democrat /*
*/   tot_pop pct_age_over65 pct_black pct_urban pct_for_born med_inc_all lnpop lnarea lninc /*
*/   charisma_dem predicted_dem rep_incumbent_cqq dem_incumbent_cqq minor_incumbent_cqq lag_demshare1 private /*
*/   borninstate agestart privatesec anycoll namedcoll ivycoll statecoll anygrad jd mba phd military /*
*/   occ0 occ1 occ2 occ3 occ4 occ5 occ6 DSpndPct DSpndPct_miss DDonaPct CQRating3 CQRating3_01 c_* /*
*/   black native asian latino mixed_gender_primary MVprim_female



sort v2 HRnumber
merge v2 HRnumber using "$IntermediateData/CosponsorsWithInfo_Clean_v2.dta"
tab _merge
drop _merge

gen int sponsor_party = party
gen byte sponsor_female = female
gen byte sponsor_tenure_run = tenure_run
gen byte sponsor_age = age
gen str2 sponsor_state_abbrev = state_abbrev
gen sponsor_state_icpsr = state_icpsr
gen int sponsor_district = district
gen byte sponsor_term_served = term_served
gen sponsor_v1_fix = v1_fix



* Identify bills cosponsored by the opposite party
gen byte cosponsor_opposite_party = (cosponsor_party==100 & sponsor_party==200) | (cosponsor_party==200 & sponsor_party==100) if sponsor_party~=. & cosponsor_party~=.
egen numb_cosponsors_opposite = sum(cosponsor_opposite_party), by(v2 HRnumber)
gen pct_cosponsors_opposite = numb_cosponsors_opposite/(numb_cosponsors)

egen tag_bill = tag(v2 HRnumber)

* calculate number of bills cosponsored
egen nbills_cosponsored = count(HRnumber), by(v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served)

egen nbills_cosponsored_opposite = sum(cosponsor_opposite_party), by(v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served)
gen pctbills_cosponsored_opposite = nbills_cosponsored_opposite/(nbills_cosponsored)


egen tag_cosponsor = tag(v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served)
egen tag_sponsor = tag(v2 sponsor_state_abbrev sponsor_district sponsor_term_served)


compress



*********************************************************************************************************

* (1) Create the RD forcing variable for sponsors and cosponsors


* (1a) Create RD forcing variables for sponsors...
gen MV1_female = MV1_democrat if sponsor_party==100 & sponsor_female==1 & mixed_gender_election==1
replace MV1_female = -MV1_democrat if sponsor_party==200 & sponsor_female==1 & mixed_gender_election==1
replace MV1_female = MV1_democrat if sponsor_party==200 & sponsor_female==0 & mixed_gender_election==1
replace MV1_female = -MV1_democrat if sponsor_party==100 & sponsor_female==0 & mixed_gender_election==1




*****UPDATE, FEBRUARY 2014
gen charisma_winner = charisma_dem if sponsor_party==100 
replace charisma_winner = -charisma_dem if sponsor_party==200 

gen predicted_winner = predicted_dem if sponsor_party==100 
replace predicted_winner = 1-predicted_dem if sponsor_party==200 

***************************************************




drop if sponsor=="Rep Herseth, Stephanie" | cosponsor=="Rep Herseth, Stephanie"

replace MV1_female = 100*MV1_female
label var MV1_female "Margin victory female sponsor"

* interactions between gender dummy and RD forcing variable
gen femaleXMV1_female=sponsor_female*MV1_female
forvalues j = 2/4{
   gen MV1_female__`j' = MV1_female^`j'
   gen femaleXMV1_female__`j' = sponsor_female*MV1_female__`j'
}


* (1b) Create RD forcing variables for cosponsors...
gen cosp_MV1_female = cosp_MV1_democrat if cosponsor_party==100 & cosponsor_female==1 & cosp_mixed_gender_election==1
replace cosp_MV1_female = -cosp_MV1_democrat if cosponsor_party==200 & cosponsor_female==1 & cosp_mixed_gender_election==1
replace cosp_MV1_female = cosp_MV1_democrat if cosponsor_party==200 & cosponsor_female==0 & cosp_mixed_gender_election==1
replace cosp_MV1_female = -cosp_MV1_democrat if cosponsor_party==100 & cosponsor_female==0 & cosp_mixed_gender_election==1

replace cosp_MV1_female = 100*cosp_MV1_female
label var cosp_MV1_female "Margin victory female cosponsor"

*****UPDATE, FEBRUARY 2014
gen cosp_charisma_winner = cosp_charisma_dem if cosponsor_party==100 
replace cosp_charisma_winner = -cosp_charisma_dem if cosponsor_party==200 

gen cosp_predicted_winner = cosp_predicted_dem if cosponsor_party==100 
replace cosp_predicted_winner = 1-cosp_predicted_dem if cosponsor_party==200 

***************************************************




* interactions between gender dummy and RD forcing variable
gen femaleXcosp_MV1_female=cosponsor_female*cosp_MV1_female
forvalues j = 2/4 {
   gen cosp_MV1_female__`j' = cosp_MV1_female^`j'
   gen femaleXcosp_MV1_female__`j' = cosponsor_female*cosp_MV1_female__`j'
}

* discretize running variable
gen  MV1_female_bins = int(MV1_female)+0.5 if MV1_female>0 & MV1_female~=. 
replace MV1_female_bins = int(MV1_female)-0.5 if MV1_female<0 & MV1_female~=.

gen  MV1_female_bins2 = 2*int(MV1_female/2)+1 if MV1_female>0 & MV1_female~=. 
replace MV1_female_bins2 = 2*int(MV1_female/2)-1 if MV1_female<0 & MV1_female~=.

gen  cosp_MV1_female_bins = int(cosp_MV1_female)+.5 if cosp_MV1_female>0 & cosp_MV1_female~=.
replace cosp_MV1_female_bins = int(cosp_MV1_female)-0.5 if cosp_MV1_female<0 & cosp_MV1_female~=.

gen  cosp_MV1_female_bins2 = 2*int(cosp_MV1_female/2)+1 if cosp_MV1_female>0 & cosp_MV1_female~=.
replace cosp_MV1_female_bins2 = 2*int(cosp_MV1_female/2)-1 if cosp_MV1_female<0 & cosp_MV1_female~=.

* these two variables may come in handy later
gen byte sponsor_democrat = sponsor_party==100
gen MV1_democratXsponsor_democrat = MV1_democrat*sponsor_democrat


* NEW OUTCOMES *

* identify bills cosponsored by women, and by women of the opposite party
gen byte cosponsor_fem = (cosponsor_female==1) if cosponsor_female~=.
egen numb_cosponsors_fem = sum(cosponsor_fem), by(v2 HRnumber)
gen pct_cosponsors_fem = numb_cosponsors_fem/(numb_cosponsors)

gen byte cosponsor_fem_opposite = (cosponsor_party==100 & sponsor_party==200 & cosponsor_female==1) | (cosponsor_party==200 & sponsor_party==100 & cosponsor_female==1) if sponsor_party~=. & cosponsor_party~=. & cosponsor_female~=.
egen numb_cosponsors_fem_opposite = sum(cosponsor_fem_opposite), by(v2 HRnumber)
gen pct_cosponsors_fem_opposite = numb_cosponsors_fem_opposite/(numb_cosponsors)

gen byte cosponsor_male_sp = (cosponsor_party==100 & sponsor_party==100 & cosponsor_female==0) | (cosponsor_party==200 & sponsor_party==200 & cosponsor_female==0) if sponsor_party~=. & cosponsor_party~=. & cosponsor_female~=.
egen numb_cosponsors_male_sp = sum(cosponsor_male_sp), by(v2 HRnumber)
gen pct_cosponsors_male_sp = numb_cosponsors_male_sp/(numb_cosponsors)

* identify cosponsored bills by women, and women of opposite party
egen nbills_cosponsored_fem = sum(sponsor_female), by(v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served)
gen pctbills_cosponsored_fem = nbills_cosponsored_fem/(nbills_cosponsored)

gen byte sponsor_female_opposite =  (cosponsor_party==100 & sponsor_party==200 & sponsor_female==1) | (cosponsor_party==200 & sponsor_party==100 & sponsor_female==1) if sponsor_party~=. & cosponsor_party~=. & sponsor_female~=.
egen nbills_cosponsored_fem_opp = sum(sponsor_female_opposite), by(v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served)
gen pctbills_cosponsored_fem_opp = nbills_cosponsored_fem_opp/(nbills_cosponsored)

* identify variance/sd/distance in DW among cosponsors
egen sd_cosp_dwnom1_nospons = sd(cosp_dwnom1), by(v2 HRnumber)
replace sd_cosp_dwnom1_nospons = . if numb_cosponsors==0
gen var_cosp_dwnom1_nospons = sd_cosp_dwnom1_nospons^2
replace var_cosp_dwnom1_nospons = . if numb_cosponsors==0
egen m_cosp_dwnom1_nospons=mean(cosp_dwnom1), by(v2 HRnumber)
replace m_cosp_dwnom1_nospons = . if numb_cosponsors==0
gen a_cosp_dwnom1=abs(cosp_dwnom1)
egen ma_cosp_dwnom1_nospons=mean(a_cosp_dwnom1), by(v2 HRnumber)
replace ma_cosp_dwnom1_nospons = . if numb_cosponsors==0
gen d_cosp_dwnom1=cosp_dwnom1-dwnom1
egen md_cosp_dwnom1_nospons=mean(d_cosp_dwnom1), by(v2 HRnumber)
replace md_cosp_dwnom1_nospons = . if numb_cosponsors==0
gen ad_cosp_dwnom1=abs(cosp_dwnom1-dwnom1)
egen mad_cosp_dwnom1_nospons=mean(d_cosp_dwnom1), by(v2 HRnumber)
replace mad_cosp_dwnom1_nospons = . if numb_cosponsors==0
drop a_cosp_dwnom1 d_cosp_dwnom1 ad_cosp_dwnom1

expand 2, gen(new)
bysort v2 HRnumber new: gen counter=_n
drop if new==1 & counter!=1 
replace cosp_dwnom1=dwnom1 if new==1
egen sd_cosp_dwnom1_spons = sd(cosp_dwnom1), by(v2 HRnumber)
replace sd_cosp_dwnom1_spons = 0 if numb_cosponsors==0
gen var_cosp_dwnom1_spons = sd_cosp_dwnom1_spons^2
replace var_cosp_dwnom1_spons = 0 if numb_cosponsors==0
egen m_cosp_dwnom1_spons=mean(cosp_dwnom1), by(v2 HRnumber)
*replace m_cosp_dwnom1_spons = . if numb_cosponsors==0
gen a_cosp_dwnom1=abs(cosp_dwnom1)
egen ma_cosp_dwnom1_spons=mean(a_cosp_dwnom1), by(v2 HRnumber)
*replace ma_cosp_dwnom1_spons = . if numb_cosponsors==0
gen d_cosp_dwnom1=cosp_dwnom1-dwnom1
egen md_cosp_dwnom1_spons=mean(d_cosp_dwnom1), by(v2 HRnumber)
replace md_cosp_dwnom1_spons = 0 if numb_cosponsors==0
gen ad_cosp_dwnom1=abs(cosp_dwnom1-dwnom1)
egen mad_cosp_dwnom1_spons=mean(d_cosp_dwnom1), by(v2 HRnumber)
replace mad_cosp_dwnom1_spons = 0 if numb_cosponsors==0
drop a_cosp_dwnom1 d_cosp_dwnom1 ad_cosp_dwnom1

drop if new==1
drop new counter

* add network measures
merge m:1 v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served using "$RawData/CentralityMeasures_all.dta", keep(master matched) keepusing(degree closeness betweenness eigenvector closeness_w betweenness_w eigenvector_w)
drop _m
foreach var of varlist degree closeness betweenness eigenvector closeness_w betweenness_w eigenvector_w {
   rename `var' cosponsor_`var'
   egen m_`var'_nospons = mean(cosponsor_`var'), by(v2 HRnumber)
   replace m_`var'_nospons=. if numb_cosponsors==0
}

expand 2, gen(new)
bysort v2 HRnumber new: gen counter=_n
drop if new==1 & counter!=1
merge m:1 v2 sponsor_state_abbrev sponsor_district sponsor_term_served using "$RawData/CentralityMeasures_all.dta", keep(master matched) keepusing(degree closeness betweenness eigenvector closeness_w betweenness_w eigenvector_w)
drop _m
foreach var of varlist degree closeness betweenness eigenvector closeness_w betweenness_w eigenvector_w {
   replace cosponsor_`var'=`var' if new==1
   egen m_`var'_spons = mean(cosponsor_`var'), by(v2 HRnumber)
*   replace m_`var'_spons=. if numb_cosponsors==0
   drop `var' cosponsor_`var'
}

drop if new==1
drop new counter



****************************************************************************************
***** UPDATE, FEBRUARY 2014: Create some additional variables measuring the characteristics of cosponsors

* identify whether cosponsor is committee chair or committee ranking members
gen byte cosponsor_leader = (cosp_comc==1) | (cosp_comr==1) if cosp_comc~=. & cosp_comr~=.
egen numb_cosponsors_leader = sum(cosponsor_leader), by(v2 HRnumber)
gen pct_cosponsors_leader = numb_cosponsors_leader/(numb_cosponsors)

* Calculate average tenure and age of cosponsors, pct of cosponsors that are rookies
egen avgtenure_cosponsors = mean(cosponsor_tenure_run), by(v2 HRnumber)
egen avgage_cosponsors = mean(cosponsor_age), by(v2 HRnumber)

gen byte cosponsor_rookie=(cosponsor_tenure_run==1) if cosponsor_tenure_run!=.
gen byte cosponsor_tenure_24 = cosponsor_tenure_run>=2 & cosponsor_tenure_run<=4 if cosponsor_tenure_run~=.
gen byte cosponsor_tenure_59 = cosponsor_tenure_run>=5 & cosponsor_tenure_run<=9 if cosponsor_tenure_run~=.
gen byte cosponsor_tenure_10plus = cosponsor_tenure_run>=10 if cosponsor_tenure_run~=.
gen diff=date_cosponsored-intro_date if date_cosponsored!=. & intro_date!=.
replace diff=. if diff<0 & diff!=.
gen byte cosponsor_samedate=(diff<=0) if diff!=.
gen byte cosponsor_samedate_opp=cosponsor_samedate*cosponsor_opposite
gen byte cosponsor_samedate_fem=cosponsor_samedate*cosponsor_female

foreach type in "rookie" "tenure_24" "tenure_59" "tenure_10plus" "samedate" "samedate_opp" "samedate_fem" {
   egen numb_cosponsors_`type' = sum(cosponsor_`type'), by(v2 HRnumber)
   gen pct_cosponsors_`type' = numb_cosponsors_`type'/(numb_cosponsors)
}

****************************************************************************************


****************************************************************************************
* MODIFIED BY DANIELE , 2013.06.04

* Identify women-friendly bills
* Three definition of Women-Friendly Bills:

* women_friend1: Based on 6 major categories (subjectively defined)
* women_friend2: identify minor topics where the fraction of female *cosponsors* is above some threshold (which threshold?)
* Women_friend3: identify minor topics in which the fraction of female *sponsors* is above the 80th percentile


* women_friend1: Based on 6 major categories (subjectively defined)
gen women_friend1=(major==2 | major==3 | major==5 | major==6 | major==12 | major==13) if major!=.
label var women_friend1 "1 if CR, HE, LA, ED, LAW, SW" 
/*
1. Macroeconomics
2. Civil Rights, Minority Issues, and Civil Liberties
3. Health
4. Agriculture
5. Labor, Employment, and Immigration
6. Education
7. Environment
8. Energy
10. Transportation
12. Law, Crime, and Family Issues
13. Social Welfare
14. Community Development and Housing Issues
15. Banking, Finance, and Domestic Commerce
16. Defense
17. Space, Science, Technology and Communications
18. Foreign Trade
19. International Affairs and Foreign Aid
20. Government Operations
21. Public Lands and Water Management
*/

* women_friend2: identify minor topics where the fraction of female *cosponsors* is above the 75th percentile
* Important: the 75th percentile means the 75th percentile of the distribution of minor topics (227 observations overall)
egen tag_minor = tag(minor)

egen x = mean(pct_cosponsors_fem) if tag_bill==1, by(minor)
egen pct_cosponsors_fem_minor = mean(x), by(v2 HRnumber)   /* assign thie number you just calculated to every observation */
sum pct_cosponsors_fem_minor if tag_minor==1, d
gen byte women_friend2 = (pct_cosponsors_fem_minor)>r(p75) if pct_cosponsors_fem_min~=.
drop x
label var women_friend2 "1 if topic where fraction of F *cosp* > 75th pctile"

* women_friend3: identify minor topics in which the fraction of female *sponsors* is above the 75th percentile
egen x = mean(sponsor_female) if tag_bill==1, by(minor)
egen pct_sponsors_fem_min = mean(x), by(v2 HRnumber)
sum pct_sponsors_fem_min if tag_minor==1, d
gen byte women_friend3 = (pct_sponsors_fem_min)>r(p75) if pct_sponsors_fem_min~=.
label var women_friend3 "1 if topic where fraction of F *sponsors* > 75th pctile"
drop x
****************************************************************************************


* identify cosponsored women-friendly bills by the opposite party, and by women
forvalues j = 1/3 {
   * identify the number of women-friendly bills cosponsored
   egen x = count(HRnumber) if women_friend`j'==1, by(v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served)
   bysort v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served: egen nwfbills_cosponsored`j'=max(x)
   replace nwfbills_cosponsored`j'=0 if nwfbills_cosponsored`j'==.
   drop x

   * calculate the percentage of women_friendly bills cosponsored that were sponsored by the opposite party, 
   * out of the total number of women-friendly bills cosponsored
   egen x = sum(cosponsor_opposite_party) if women_friend`j'==1, by(v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served)
   bysort v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served: egen nwfbills_cosponsored_opposite`j'=max(x)
   replace nwfbills_cosponsored_opposite`j'=0 if nwfbills_cosponsored_opposite`j'==.
   drop x
   gen pctwfbills_cosponsored_opposite`j' = nwfbills_cosponsored_opposite`j'/(nwfbills_cosponsored`j')

   * calculate the percentage of women_friendly bills cosponsored that were sponsored by a female, 
   * out of the total number of women-friendly bills cosponsored
   egen x = sum(sponsor_female) if women_friend`j'==1, by(v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served)
   bysort v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served: egen nwfbills_cosponsored_female`j'=max(x)
   replace nwfbills_cosponsored_female`j'=0 if nwfbills_cosponsored_female`j'==.
   drop x
   gen pctwfbills_cosponsored_female`j' = nwfbills_cosponsored_female`j'/(nwfbills_cosponsored`j')

   * calculate the percentage of women_friendly bills cosponsored that were sponsored by a female opposite, 
   * out of the total number of women-friendly bills cosponsored
   egen x = sum(sponsor_female_opp) if women_friend`j'==1, by(v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served)
   bysort v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served: egen nwfbills_cosponsored_femopp`j'=max(x)
   replace nwfbills_cosponsored_femopp`j'=0 if nwfbills_cosponsored_femopp`j'==.
   drop x
   gen pctwfbills_cosponsored_femopp`j' = nwfbills_cosponsored_femopp`j'/(nwfbills_cosponsored`j')
}
****************************************************************************************

****************************************************************************************
*** UPDATE, FEBRUARY 2014: Some measures of network
*** Can probably be done a lot better

* Calculate the number of different cosponsors had per congress
* This is done in a bit of a roundabout way
sort sponsor v2 sponsor_state_abbrev sponsor_district sponsor_term_served cosponsor_v1_fix
by sponsor v2 sponsor_state_abbrev sponsor_district sponsor_term_served: gen byte x = 1 if cosponsor_v1_fix~=cosponsor_v1_fix[_n-1] & cosponsor_v1_fix~=.
sort sponsor v2 sponsor_state_abbrev sponsor_district sponsor_term_served x cosponsor_v1_fix
by sponsor v2 sponsor_state_abbrev sponsor_district sponsor_term_served: replace x = x[_n-1]+1 if x~=. & _n>=2
sort sponsor v2 sponsor_state_abbrev sponsor_district sponsor_term_served cosponsor_v1_fix x
by sponsor v2 sponsor_state_abbrev sponsor_district sponsor_term_served: replace x = x[_n-1] if x==. & x[_n-1]~=.
egen sponsor_ndistinct_cosponsors = max(x), by(v2 sponsor_state_abbrev sponsor_district sponsor_term_served)

egen sponsor_total_cosponsors = count(cosponsor_v1_fix), by(v2 sponsor_state_abbrev sponsor_district sponsor_term_served)
gen sponsor_distinctbytotal = sponsor_ndistinct_cosponsors/sponsor_total_cosponsors

gen byte cons=1
egen nbills_with_cosponsor = count(cons), by(sponsor v2 sponsor_state_abbrev sponsor_district sponsor_term_served cosponsor_v1_fix)
gen cosponsor_share_sq= (nbills_with_cosponsor/sponsor_total_cosponsors)^2
egen tag_pair = tag(sponsor v2 sponsor_state_abbrev sponsor_district sponsor_term_served cosponsor_v1_fix)
egen y = sum(cosponsor_share_sq) if tag_pair==1, by(sponsor v2 sponsor_state_abbrev sponsor_district sponsor_term_served)
egen sponsor_herfindahl = mean(y), by(sponsor v2 sponsor_state_abbrev sponsor_district sponsor_term_served)

drop x y cons tag_pair cosponsor_share_sq nbills_with_cosponsor
****************************************************************************************





****************************************************************************************
*** UPDATE, FEBRUARY 2014: Identify which bills made it to the floor
* Identify which bills got to the floor
preserve
drop _all
use "$RawData/votes_rollcall_analysis_102-111_clean.dta", clear
egen tag_bill2 = tag(v2 b_HRnumber)
keep if tag_bill2==1
keep v2 b_HRnumber
ren b_HRnumber HRnumber
gen byte reached_floor = 1
sort v2 HRnumber
save "$IntermediateData/reached_floor_102-111_clean.dta", replace 
restore

sort v2 HRnumber
merge v2 HRnumber using "$IntermediateData/reached_floor_102-111_clean.dta"
tab _merge
drop _merge
replace reached_floor = 0 if reached_floor==. & v2>=102
****************************************************************************************



* save datasets
preserve
keep v2 sponsor_state_abbrev sponsor_district sponsor_term_served cosponsor_state_abbrev cosponsor_district cosponsor_term_served
egen tag_sponsor_cosponsor=tag(v2 sponsor_state_abbrev sponsor_district sponsor_term_served cosponsor_state_abbrev cosponsor_district cosponsor_term_served)
keep if tag_sponsor_cosponsor==1
foreach var of varlist cosponsor_state_abbrev cosponsor_district cosponsor_term_served {
	rename `var' `var'2
}
foreach var of varlist sponsor_state_abbrev sponsor_district sponsor_term_served {
	rename `var' co`var'
}
compress
drop tag_sponsor_cosponsor
save "$IntermediateData/cosponsors_names_101-111_clean.dta", replace
restore


preserve
merge m:m v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served using "$AnalysisData/cosponsors_names_101-111_clean.dta", keep(master matched)
gen cross=sponsor_state_abbrev==cosponsor_state_abbrev2 & sponsor_district==cosponsor_district2 & sponsor_term_served==cosponsor_term_served2
gen cross_opp=sponsor_state_abbrev==cosponsor_state_abbrev2 & sponsor_district==cosponsor_district2 & sponsor_term_served==cosponsor_term_served2 & [sponsor_party==100 & cosponsor_party==200 | sponsor_party==200 & cosponsor_party==100]
egen tag_cosponsor2=tag(v2 HRnumber cosponsor_state_abbrev cosponsor_district cosponsor_term_served)
replace cross=cross*tag_cosponsor2
bysort v2 HRnumber sponsor_state_abbrev sponsor_district sponsor_term_served: egen numb_cosponsors_cross=sum(cross)
replace numb_cosponsors_cross=. if  numb_cosponsors==.
replace cross_opp=cross_opp*tag_cosponsor2
bysort v2 HRnumber sponsor_state_abbrev sponsor_district sponsor_term_served: egen numb_cosponsors_cross_opp=sum(cross_opp)
replace numb_cosponsors_cross_opp=. if  numb_cosponsors==.
drop tag_bill tag_cosponsor2
egen tag_bill = tag(v2 HRnumber)
keep if tag_bill==1
keep v2 HRnumber /*
*/   name_clean sponsor sponsor_female sponsor_tenure_run sponsor_age sponsor_party /*
*/   sponsor_state_abbrev sponsor_state_icpsr sponsor_district sponsor_term_served sponsor_v1_fix /*
*/   comc comr dwnom1 dwnom2 pus /*
*/   borninstate agestart privatesec anycoll namedcoll ivycoll statecoll anygrad jd mba phd military /*
*/   occ0 occ1 occ2 occ3 occ4 occ5 occ6 DSpndPct DSpndPct_miss DDonaPct CQRating3 CQRating3_01  /*
*/   pct_cosponsors_opposite numb_cosponsors_opposite numb_cosponsors /*
*/   pct_cosponsors_leader numb_cosponsors_leader /*
*/   avgtenure_cosponsors avgage_cosponsors /*
*/   numb_cosponsors_rookie numb_cosponsors_tenure_24 numb_cosponsors_tenure_59 numb_cosponsors_tenure_10plus numb_cosponsors_samedate /*
*/   pct_cosponsors_rookie pct_cosponsors_tenure_24 pct_cosponsors_tenure_59 pct_cosponsors_tenure_10plus pct_cosponsors_samedate /*
*/   numb_cosponsors_fem pct_cosponsors_fem numb_cosponsors_fem_opposite pct_cosponsors_fem_opposite numb_cosponsors_male_sp pct_cosponsors_male_sp /*
*/   numb_cosponsors_samedate_opp numb_cosponsors_samedate_fem pct_cosponsors_samedate_opp pct_cosponsors_samedate_fem /*
*/   MV1_female MV1_female__2 MV1_female__3 MV1_female__4 MV1_female_bins MV1_female_bins2 /*
*/   femaleXMV1_female femaleXMV1_female__2 femaleXMV1_female__3 femaleXMV1_female__4 /*
*/   mixed_gender_election republican democratic minor1_vote /*
*/   losing_candidate_clean_cqq losing_candidate_gender sample1 demshare1 repshare1 MV1_democrat /*
*/   intro_date numb_committees_master plaw_master lma_date iter_length mult commem house_* /*
*/   major minor passh passs plaw_cbp plawdate reached_floor /*
*/   tot_pop pct_age_over65 pct_black pct_urban pct_for_born med_inc_all lninc lnpop lnarea /*
*/   charisma_winner charisma_dem predicted_winner predicted_dem/*
*/   rep_incumbent_cqq dem_incumbent_cqq minor_incumbent_cqq lag_demshare1 /*
*/   women_friend1 women_friend2 women_friend3 private c_* numb_cosponsors_cross numb_cosponsors_cross_opp /*
*/   black native asian latino sd_cosp_* var_cosp_* m_cosp_* ma_cosp_* md_cosp_* mad_cosp_* /*
*/   m_degree_* m_betweenness_* m_eigenvector_* m_closeness_w_* mixed_gender_primary MVprim_female

egen sponsor_v1_flex=group(sponsor_v1_fix v2)

save "$IntermediateData/bills_analysis_101-111_clean.dta", replace     // This smaller sample keeps just one observation per bill, allows to quickly reproduce the regression results
restore


preserve
tab minor, gen(minor_)
foreach var of varlist house_* minor_* {
   bysort v2 cosponsor_state_abbrev cosponsor_district cosponsor_term_served: egen `var'_m=mean(`var')
   drop `var'
}
keep if tag_cosponsor==1
keep cosponsor v2 cosponsor_state_abbrev cosponsor_state_icpsr cosponsor_district cosponsor_term_served cosponsor_v1_fix /*
*/   cosp_mixed_gender_election cosp_MV1_democrat/*
*/   cosponsor_age cosponsor_female cosponsor_tenure_run cosponsor_party /* 
*/   cosp_name_clean cosp_pus cosp_comc cosp_comr  cosp_dwnom1 cosp_dwnom2 /*
*/   cosp_losing_candidate_clean_cqq cosp_losing_candidate_gender cosp_mixed_gender_election /*
*/   cosp_sample1 cosp_demshare1 cosp_repshare1 cosp_MV1_democrat /*
*/   cosp_tot_pop cosp_pct_age_over65 cosp_pct_black cosp_pct_urban cosp_pct_for_born cosp_med_inc_all /*
*/   cosp_lninc cosp_lnarea cosp_lnpop /*
*/   cosp_charisma_winner cosp_charisma_dem cosp_predicted_winner cosp_predicted_dem /*
*/   cosp_rep_incumbent_cqq cosp_dem_incumbent_cqq cosp_minor_incumbent_cqq cosp_lag_demshare1 /*
*/   cosp_borninstate cosp_agestart cosp_privatesec cosp_anycoll cosp_namedcoll cosp_ivycoll cosp_statecoll cosp_anygrad cosp_jd cosp_mba cosp_phd /*
*/   cosp_military cosp_occ0 cosp_occ1 cosp_occ2 cosp_occ3 cosp_occ4 cosp_occ5 cosp_occ6 cosp_DSpndPct cosp_DSpndPct_miss cosp_DDonaPct cosp_CQRating3 cosp_CQRating3_01 /*
*/   nbills_cosponsored nbills_cosponsored_opposite pctbills_cosponsored_opposite /*
*/   nbills_cosponsored_fem pctbills_cosponsored_fem nbills_cosponsored_fem_opp pctbills_cosponsored_fem_opp /*
*/   cosp_MV1_female femaleXcosp_MV1_female cosp_MV1_female__2 femaleXcosp_MV1_female__2 /*
*/   cosp_MV1_female__3 femaleXcosp_MV1_female__3 cosp_MV1_female__4 femaleXcosp_MV1_female__4 /*
*/   cosp_MV1_female_bins cosp_MV1_female_bins2  /*
*/   nwfbills_cosponsored* nwfbills_cosponsored_opposite* pctwfbills_cosponsored_opposite* /*
*/   nwfbills_cosponsored_femopp* pctwfbills_cosponsored_femopp* /*
*/   nwfbills_cosponsored_female* pctwfbills_cosponsored_female* cosp_c_* cosp_black cosp_native cosp_asian cosp_latino /* 
*/   house_* minor_* 

egen cosponsor_v1_flex=group(cosponsor_v1_fix v2)

save "$IntermediateData/cosponsors_analysis_101-111_clean.dta", replace    // keep this smaller dataset that can allow quick reproduction of the regressions in table 2
restore

preserve
keep if tag_sponsor==1
keep v2 sponsor_state_abbrev sponsor_district sponsor_term_served  /*
*/   name_clean sponsor sponsor_female sponsor_tenure_run sponsor_age sponsor_party /*
*/   sponsor_state_icpsr sponsor_v1_fix /*
*/   comc comr dwnom1 dwnom2 pus /*
*/   borninstate agestart privatesec anycoll namedcoll ivycoll statecoll anygrad jd mba phd military /*
*/   occ0 occ1 occ2 occ3 occ4 occ5 occ6 DSpndPct DSpndPct_miss DDonaPct CQRating3 CQRating3_01  /*
*/   MV1_female MV1_female__2 MV1_female__3 MV1_female__4 MV1_female_bins MV1_female_bins2 /*
*/   femaleXMV1_female femaleXMV1_female__2 femaleXMV1_female__3 femaleXMV1_female__4 /*
*/   mixed_gender_election republican democratic minor1_vote /*
*/   losing_candidate_clean_cqq losing_candidate_gender sample1 demshare1 repshare1 MV1_democrat /*
*/   tot_pop pct_age_over65 pct_black pct_urban pct_for_born med_inc_all lninc lnarea lnpop /*
*/   charisma_winner charisma_dem predicted_winner predicted_dem/*
*/   rep_incumbent_cqq dem_incumbent_cqq minor_incumbent_cqq lag_demshare1 /*
*/   sponsor_ndistinct_cosponsors sponsor_total_cosponsors sponsor_distinctbytotal sponsor_herfindahl c_* /*
*/   black native asian latino mixed_gender_primary MVprim_female

egen sponsor_v1_flex=group(sponsor_v1_fix v2)

save "$IntermediateData/sponsors_analysis_101-111_clean.dta", replace    // keep this smaller dataset that can allow quick reproduction of the regressions in table 2
restore



log close